

standardisierte verkehrstechnische Aufgabenstellung

SVTA

1 Zielsetzung

Verkehrstechnische Aufgabenstellungen für Lichtsignalanlagen LSA werden üblicherweise in Textform abgefasst. Auf diese Weise lässt sich jede gewünschte Reaktion auf alle denkbaren Verkehrssituationen exakt beschreiben. Die verstärkte Nutzung dieser hohen Flexibilität führt im gleichen Zug zu immer höherer Unübersichtlichkeit.

Ziel der standardisierten verkehrstechnischen Aufgabenstellung SVTA ist es, diese Unübersichtlichkeit zu minimieren. Gleichzeitig führen die vorgegebenen Syntaxregeln zu einer klaren Struktur, die es dem Programmierer der LSA-Steuerung erlaubt, die Anforderungen mit wenigen Blicken zu erfassen. Ebenso vorstellbar ist es, dass die SVTA von der hier beschriebenen Textform automatisch in ein LSA-Steuerprogramm umgewandelt wird. Die maschinenlesbare Form der SVTA eröffnet neue Möglichkeiten hinsichtlich weiterer Vereinfachung und schnellerer Umsetzung von LSA-Steuerungen.

Die folgenden Kapitel beschreiben den genauen Aufbau der SVTA. Es handelt sich bei dieser Beschreibung weniger um eine vollständige Sprachdefinition, sondern um eine Beispielsammlung, die es dem Praktiker ermöglicht, einen raschen Einstieg zu finden.

2 Form

Für die SVTA gilt Textform. Auch wenn sich die Informationen der SVTA ohne weiteres in Tabellen speichern lassen, wird aus Gründen der leichteren Gliederung und besseren Lesbarkeit die textliche Darstellung bevorzugt. Die einzelnen Zeilen des Textes enthalten Regeln für den Ablauf der LSA-Steuerung. Diese Regeln treffen Festlegungen hinsichtlich Zwischenzeiten, Gelbzeiten, Höchstfreigabezeiten und vieler anderer steuerungstechnischer Details.

Die SVTA wird immer von oben nach unten gelesen. Das bedeutet, dass einmal festgelegte Aspekte der LSA-Steuerung weiter unten im Text näher präzisiert oder auch abgeändert werden können. So lässt sich beispielsweise eine pauschal für alle Signalgruppen festgelegte Mindestgrünzeit für einzelne dieser Signalgruppen abändern.

3 Regeln

Im Wesentlichen handelt es sich bei der SVTA um eine Zusammenstellung von Regeln, an die sich die LSA-Steuerung zu halten hat. Jede dieser Regeln beginnt mit einem Regelnamen. Dieser besteht aus einem oder zwei Buchstaben. Meist folgen darauf eine Signalgruppennummer, ein Operator und ein Zahlenwert. Den Aufbau der einzelnen Regelarten zeigen die folgenden Abschnitte.

3.1 Signalprogramm

Innerhalb einer SVTA können mehrere Signalprogramme beschrieben werden. Die p-Regelzeile steht jeweils zu Beginn des Textabschnitts, der signalprogrammsspezifische Regeln enthält.

- p1** Hier beginnen die Regeln für Signalprogramm Nummer 1.
- p2-5** Dgl. für die Signalprogramme 2 bis 5.
- p7,9,12-16** Dgl. für die Signalprogramme 7, 9, 12, 13, 14, 15 und 16.
- p** Von dieser Zeile an gelten die Regeln für alle Signalprogramme.

3.2 Umlaufzeit

Mittels tu-Regelzeile wird die Umlaufzeit angegeben. Sie ist insbesondere für koordinierte LSA-Steuerung und für die Interpretation der Sollgrünzeit von Bedeutung.

- tu=50** Es gilt eine Umlaufzeit von 50 Sekunden.

3.3 Gelbzeit

Die Übergangszeit vom Freigabezustand zum Sperrzustand einer Signalgruppe (hier der Einfachheit halber unabhängig von ihrem tatsächlichen Signalbild als Gelbzeit bezeichnet) wird mit der ge-Regelzeile festgelegt.

- ge1=3** Drei Sekunden Gelbzeit für die Signalgruppe 1.
- ge2-4,6=4** Vier Sekunden Gelbzeit für die Signalgruppen 2, 3, 4 und 6.

3.4 Rotgelbzeit

Die Übergangszeit für den umgekehrten Signalzustandswechsel wird in gleicher Weise wie die Gelbzeit festgelegt. Der Regelname hierfür lautet "rg".

- rg8= 2** Zwei Sekunden Rotgelbzeit für die Signalgruppe 8.
- rg3,6-7= 1** Eine Sekunde Rotgelbzeit für die Signalgruppen 3 und 6 bis 7.

3.5 Mindestrotzeit

Die r-Regelzeile legt Mindestsperrzeiten (Mindestrotzeiten) fest. Übergangszeiten sind nicht enthalten. Das mathematische Zeichen "≥" für "mindestens" bzw. "größer oder gleich" wird – wie in der Datenverarbeitung üblich – als ">=" geschrieben.

- r3-6>=2** Für Signalgruppe 3 bis 6 gilt eine Mindestrotzeit von 2 Sekunden.
- r7>8** Die Rotzeit für Signalgruppe 7 ist immer größer als 8 Sekunden; somit gilt hier eine Mindestrotzeit von 9 Sekunden.

3.6 Höchstwartezeit

Höchstwartezeiten (Höchstrotzeiten) werden ebenfalls per r-Regelzeile festgelegt. Bei Signalgruppen, die nur auf Anforderung freigegeben werden, zählt die Wartezeit erst ab dem Zeitpunkt der Anforderung.

- r4<=120** Signalgruppe 4 soll nicht länger als 120s auf die Freigabe warten.
- r5-7,11<90** Signalgruppen 5 bis 7 und 11 sollen weniger als 90s warten.

3.7 Mindestgrünzeit

Mit g-Regelzeilen werden Mindestfreigabezeiten (Mindestgrünzeiten) beschrieben. Die sehr mächtige g-Regelzeile legt auch viele andere Steuerungsparameter fest. Deshalb kommt es hier besonders darauf an, den richtigen Regeloperator zu verwenden. Für die Mindestgrünzeit ist es das mathematische Zeichen für "größer oder gleich".

- g16>=10** Zehn Sekunden Mindestgrünzeit für Signalgruppe 16.
g1-6,8>=5 Fünf Sekunden Mindestgrünzeit für die Signalgruppen 1 bis 6 und 8.
g1,3,5>6 Grünzeit für die Signalgruppen 1, 3 und 5 größer als 6 Sekunden;
 somit gilt hier eine Mindestgrünzeit von 7 Sekunden.

3.8 Höchstgrünzeit

Ebenfalls per g-Regelzeile wird die Höchstfreigabezeit (Höchstgrünzeit) angegeben. Diese soll "kleiner oder gleich" dem angegebenen Zahlenwert sein. Deshalb wird hier das entsprechende mathematische Zeichen " \leq " verwendet.

- g3<=9** Neun Sekunden Höchstgrünzeit für Signalgruppe 3.
g2,4,6-8<=9 Dgl. für Signalgruppen 2, 4 und 6 bis 8.
g1-3,5-7<10 Ebenfalls neun Sekunden Höchstgrünzeit für Sg. 1, 2, 3, 5, 6 und 7.

3.9 Exakte Grünzeit

An Stelle einer Mindest- oder Höchstgrünzeit kann auch die genaue Freigabezeit (genaue Grünzeit) festgelegt werden. Der betreffende Regeloperator ist das "="-Zeichen.

- g2=5** Exakt fünf Sekunden Freigabezeit für Signalgruppe 2.
g3-4,6=5 Dgl. für die Signalgruppen 3 bis 4 und 6.

3.10 Sollgrünzeit

Für jede Signalgruppe kann eine Sollfreigabezeit (Sollgrünzeit) vorgegeben werden. Ziel der LSA-Steuerung muss es dann sein, in jedem Umlauf diese Sollfreigabezeit zu erreichen. Bei verkehrsabhängigen Eingriffen kann das häufig nicht garantiert werden. Hier gilt es, die in einem Umlauf versäumte Sollfreigabezeit im nächsten (oder übernächsten) Umlauf nachzuholen. Auf diese Weise wird gewährleistet, dass sich ein durch ungünstige Häufung von ÖV-Eingriffen entstandener IV-Stau rasch wieder abbaut.

Verwendung findet hier das mathematische Zeichen " $\stackrel{!}{=}$ " für "soll gleich".

- g4! $\stackrel{!}{=}$ 20** 20 Sekunden Sollgrünzeit für Signalgruppe 4.
g5,6-9! $\stackrel{!}{=}$ 20 Dgl. für die Signalgruppen 5 und 6 bis 9.

3.11 Zwischenzeit

Für die Zwischenzeiten (Schutzzeiten) sieht der Syntax der SVTA Regeln mit dem Namen "zz" vor. Wie auch bei den weiter unten beschriebenen Regeln "gg" und "rr", deren Namen ebenfalls aus zwei gleichen Buchstaben bestehen, gilt hier eine andere Art der Indizierung. Zwischenzeiten beziehen sich nicht auf einzelne Signalgruppen, sondern immer paarweise auf zwei. Daraus folgend müssen links des Regeloperators ("=") grundsätzlich zwei Signalgruppennummern angegeben werden. Um den Schreibaufwand in Grenzen zu halten, ist auch die Matrixform möglich. Dabei wird links des "=" nur eine Signalgruppe angegeben. Rechts des "=" folgt dann eine Aufzählung der Zwischenzeiten für die abhängigen Signalgruppen (beginnend bei Signalgruppe 1).

Signalgruppen, zwischen denen eine Zwischenzeit definiert wurde, gelten als "feindlich", das heißt, sie dürfen zu keinem Zeitpunkt gleichzeitig freigegeben sein. Bei bedingt verträglichen Verkehrsströmen (z.B. Linksabbieger des Gegenverkehrs oder parallel laufende Fußgänger) müssen in der Regel Zwischenzeiten eingehalten werden, ohne dass eine Feindlichkeit besteht (siehe deutsche RiLSA 2.6.7 unten). Diese Zwischenzeiten sind mit einem Stern zu kennzeichnen.

- zz1,2=5** Fünf Sekunden Zwischenzeit von Signalgruppe 1 zu Signalgruppe 2.

- zz2=5, ,6, ,7** Zwischenzeit von Signalgruppe 2 zu Signalgruppe 1: 5 Sekunden,
von Signalgruppe 2 zu Signalgruppe 3: 6 Sekunden und
von Signalgruppe 2 zu Signalgruppe 5: 7 Sekunden.
- zz3,2=*4** Vier Sekunden Zwischenzeit, aber keine Feindlichkeit.
- zz3,4=-2** Negative Zwischenzeit (Grün-Grün-Überlappung).

3.12 Versatz Beginn

Für das gleichzeitige Freigeben von Signalgruppen kann eine Versatzzeit angegeben werden. So wird Seitens der verkehrstechnischen Aufgabenstellung Einfluss genommen auf Elemente des Phasenübergangs. Auf diese Art lässt sich beispielsweise das verzögerte Zuschalten einer Fahrzeugsignalgruppe erreichen.

Es gilt auch hier die bereits im Abschnitt Zwischenzeit beschriebene Indizierungsart. Werden für die Beziehung zwischen zwei Signalgruppen zwei Ungleichungen angegeben, lässt sich damit ein Zeitbereich für das Zuschalten festlegen. Sind die betreffenden Ungleichungen so gewählt, dass sie nie erfüllbar sind, weil sie sich gegenseitig ausschließen, bedeutet das, dass nur eine von beiden eingehalten werden muss. Damit kann ein Zeitbereich für das verzögerte Zuschalten ausgeschlossen werden.

Zu beachten ist, dass die alleinige Angabe einer Versatzzeit nicht sicherstellt, dass von den betroffenen Signalgruppen wirklich alle freigegeben werden. Die Versatzzeit gilt nur für den Fall, dass die gemeinsame Freigabe (oder das Hinzuschalten) aus anderen Gründen erfolgt, zum Beispiel durch eine Implikation.

- gg5,6=2** Grünbeginn der Signalgruppe 6 immer exakt 2s nach dem der Sg. 5.
- gg5,6>=2** Dgl., aber Grünbeginn mindestens zwei Sekunden später.
- gg5,6>2** Dgl., aber Grünbeginn mehr als zwei Sekunden später.
- gg5,6<=2** Dgl., aber Grünbeginn maximal zwei Sekunden später.
- gg5,6<2** Dgl., aber Grünbeginn weniger als zwei Sekunden später.
- gg8=,4,5** Grünbeginn der Signalgruppe 2 exakt 4s nach dem der Signalgruppe 8,
Grünbeginn der Signalgruppe 3 exakt 5s nach dem der Signalgruppe 8.

3.13 Versatz Ende

Für diese Regelart gilt das Gleiche wie für die Regel "Versatz Beginn". Das Grüne mehrerer Signalgruppen kann zeitlich auf einander abgestimmt werden. So lässt sich beispielsweise die Nachblinkzeit eines Schutzblinkers festlegen.

- rr5,6=2** Grünende der Signalgruppe 6 immer exakt 2s nach dem der Sg. 5.
- rr5,6>=2** Dgl., aber Grünende mindestens zwei Sekunden später.
- rr5,6>2** Dgl., aber Grünende mehr als zwei Sekunden später.
- rr5,6<=2** Dgl., aber Grünende maximal zwei Sekunden später.
- rr5,6<2** Dgl., aber Grünende weniger als zwei Sekunden später.
- rr8=,4,5** Grünende der Signalgruppe 2 exakt 4s nach dem der Signalgruppe 8,
Grünende der Signalgruppe 3 exakt 5s nach dem der Signalgruppe 8.

3.14 Implikation

Per Implikation kann der Freigabezustand einer Signalgruppe an eine andere weitergegeben werden. Versatzbedingungen, die zwischen diesen beiden Signalgruppen definiert wurden, müssen von der LSA-Steuerung eingehalten werden.

Auch die logische Verknüpfung mehrerer Signalgruppen ist möglich. Die Operatoren "&" für "und zugleich" und "|" für "oder auch" finden hierbei Verwendung. Innerhalb des logischen (booleschen) Terms werden Klammern eingesetzt, um die Abarbeitungsreihenfolge der Operationen festzulegen.

- g1:=3** Die Signalgruppe 1 wird immer dann freigegeben, wenn auch die Signalgruppe 3 Frei erhält.
- g1:=3&4** Dgl., aber wenn die Signalgruppen 3 und 4 zugleich freigegeben sind.
- g1:=3|4** Dgl., aber wenn die Signalgruppe 3 oder auch 4 freigegeben ist.
- g1:=3&(4|5)** Komplexerer logischer Term. Beispielsweise:
Signalgruppe 3 ist Fahrzeug-Hauptrichtung,
Signalgruppe 1 ist Diagonalblinker, Signalgruppe 4 Gegenverkehr und
Signalgruppe 5 der parallel zum Gegenverkehr laufende Fußgänger.

3.15 Welle

Die w-Regelzeile erlaubt Angaben hinsichtlich der Koordinierung. Für jede Signalgruppe kann festgelegt werden, zu welchen Umlaufsekunden sie Grün erhalten soll. Die einzelnen Zeitbereiche können jeweils mit Prioritätsangaben versehen werden. Durch diese Information ist eine Abwägung bei konkurrierenden Koordinierungsanforderungen möglich. Für die Priorität gilt ein Wertebereich von 1 bis 100. Dabei bedeuten 1 sehr niedrige und 100 besonders hohe Priorität.

- w4=11-20!10** Die Signalgruppe 4 soll in den Umlaufsekunden 11 bis 20 Grün erhalten. Die Priorität hierfür beträgt 10.
- w5=33!50,34-38!10** Die Signalgruppe 5 soll von Umlaufsekunde 33 bis 38 Grün erhalten, wichtig hierbei ist die Sekunde 33.
- w6=21-25!10,41-45!20** Die Signalgruppe 6 soll innerhalb des Umlaufs zweimal Grün erhalten; dabei ist der zweite Bereich wichtiger.

3.16 Ereignis

Reaktionen auf externe und interne Ereignisse werden mit e-Regeln beschrieben. Die SVTA macht diesbezüglich keinen Unterschied zwischen Detektoren und Signalgruppen. Reagiert werden kann folglich sowohl auf den Zustand einer Signalgruppe als auch auf das Auslösen eines Detektors. Dadurch lässt sich beispielsweise die versetzte Freigabe hintereinander liegender Radfurten sehr einfach beschreiben: Das Grün der ersten Furt fordert die Freigabe der zweiten an.

Erkauft wird dieser Vorteil mit einer Einschränkung hinsichtlich des Zahlenbereichs für Signalgruppen- und Detektornummern. Kein Detektor darf die gleiche Nummer besitzen wie eine Signalgruppe. Das ist durch entsprechende Wahl der Wertebereiche sicherzustellen.

Dem Regelnamen "e" folgen vor dem Regeloperator ("=") bis zu vier Indizes. Die ersten beiden sind Pflichtangaben, dritter und vierter Index sind wahlfrei. Die Bedeutung der Indizes im Einzelnen:

1. Ziel: Signalgruppe, deren Freigabe über das Ereignis angefordert wird.
2. Anmelder: Detektor (oder Signalgruppe), die das Ereignis startet.
3. Zähler: Maximale Anzahl gleichzeitig gespeicherter Anmeldungen. Damit lassen sich ÖV-Eingriffe mit Einzählen/Auszählen beschreiben. Wird kein vierter Index (Abmelder) angegeben, so bezeichnet dieser Wert die tolerierte Überfahrzeit. Fahrzeuge, die innerhalb der Überfahrzeit nach Grünende den Signalquerschnitt passieren, lösen keine erneute Anmeldung aus.
4. Abmelder: Detektor (oder Signalgruppe), die das Ereignis beendet. Wird die Nummer negativ angegeben, soll auf den umgekehrten Zustandswechsel reagiert werden.

Für die nach dem Regeloperator aufzählbaren Zeitbereiche gilt das bereits bei den Wellen Beschriebene: Zeitbereiche mit Prioritätsangabe legen fest, wann die Ziel-

signalgruppe (1. Index) Frei erhalten soll. Im Vergleich zu den Wellen existieren jedoch zwei Besonderheiten:

- Die angegebenen Zeitbereiche beziehen sich nicht auf die Umlaufsekunde, sondern auf den Zeitpunkt, ab dem die Anmeldung ausgelöst wurde. Angegeben wird also immer die seit der Anmeldung verstrichene Zeit.
- Das Ende des letzten angegebenen Zeitbereichs darf entfallen. Das bedeutet, dass die Anmeldung so lange nicht gelöscht wird, bis die Zielsignalgruppe wirklich abgemeldet wurde (bei fehlendem Abmelder, bis sie Grün erhalten hat).

e1,8=0-!10	Signalgruppe 1 soll mit Detektor 8 angefordert werden, gültig ab sofort und mit Priorität 10.
e1,8=20-!10	Dgl., aber mit 20s Verzögerung.
e1,8=20-40!10	Dgl., aber mit Stornierung der Anforderung nach 40s.
e1,8=0-20!10,21-!30	Erhöhung der Priorität nach 20 Sekunden.
e1,8,0,-8=0-!10	Anmeldung bleibt so lange, bis Detektor 8 gelöscht ist.
e1,8,3=0-!10	Drei Sekunden Überfahrzeit.
e1,8,2,9=10-50!20	Bus-Anforderung. Anmelde-det. 8, Abmelde-det. 9, max. zwei Busse gleichzeitig werden gezählt, geplante Fahrzeit bis zur LSA 10 Sekunden, Zwangsabmeldung nach 50 Sekunden.

3.17 Namen

Zur Verbesserung der Lesbarkeit können freie Namen (Platzhalter) vergeben werden. Zur Unterscheidung von Regelnamen beginnen Platzhalternamen grundsätzlich mit einem oder mehreren Großbuchstaben.

K1=4	Die Signalgruppe 4 heißt ab jetzt auch "K1".
Bus1=5	Die Signalgruppe 5 heißt ab jetzt auch "Bus1"
WelleHR=11-20!10	Dieser Wellenbereich heißt ab jetzt "WelleHR"
g Bus1 >=8	Entspricht "g5>=8".
w K1 = WelleHR	Entspricht "w4=11-20!10".

3.18 Kommentare

Jede Textstelle innerhalb der SVTA kann mit einem Kommentar versehen werden. Kommentare beginnen mit doppelten Schrägstrichen ("//").


// Lichtsignalanlage Badstraße/Turmstraße

g1-6>=5 // grundsätzlich 5s Mindestgrünzeit

g4>=10 // für Signalgruppe 4 jedoch 10s Mindestgrünzeit

4 Beispiel

Zur Verdeutlichung der SVTA-Struktur folgt ein Beispiel. Bei der beschriebenen LSA handelt es sich um die Signalisierung einer Einmündung. Es wurden Details zur Grünen Welle und zur Fahrzeuganforderung vorgegeben.



```
// Einmündung Bahnhofstraße

// Namen
K1= 1 // Fahrzeug-Signal HR West
K2= 2 // Fahrzeug-Signal NR
K3= 3 // Fahrzeug-Signal HR Ost
Fg= 4 // Fußgänger über NR
B1= 5 // Schutzblinker für Fg
Det= 21 // Detektor für NR

// Übergangszeiten
ge K1-K3= 3 // 3s Gelbzeit
rg K1-K3= 1 // 1s Rotgelbzeit

// Mindestgrün- und -rotzeiten
g 1-4>= 5 // Standard-Mindestgrünzeit
g K1,K3>= 8 // Fahrzeug-Hauptrichtung
r 1-4>= 2 // Mindestrotzeit

// Zwischenzeiten
// K1 K2 K3 Fg
zz K1= , 5,*6,*6
zz K2= 6, , 6, 4
zz K3= , 6, ,*6
zz Fg= , 7

// Versatzzeiten
gg K3,K1>= 1 // wg. Frühstart der LA
gg Fg>= 0,,0 // Fußgänger-Vorsprung
rr Fg,B1>= 7 // Nachblinkzeit

// Implikation
g B1:= Fg // Schutzblinker für Fg

// Ereignis
e K2,Det,2= 0-!10 // Anforderung NR
// per Detektor 21, Überfahrzeit 2s

p1 // Signalprogramm 1
tu= 50 // Umlaufzeit
g K1,K3!= 25 // 25s Sollgrün je Umlauf
g Fg!= 1 // wirksam wird Mindestgrünzeit
w K1= 10-25!20 // Welle für K1

p2 // Signalprogramm 2
tu= 75 // Umlaufzeit
g K1,K3!= 30 // 30s Sollgrün je Umlauf
g Fg!= 1 // wirksam wird Mindestgrünzeit
w K1= 40-60!20 // Welle für Signal K1
zz K2,Fg= 6 // wegen Überstauung
```